



SPECTRAVIDEO COMPUTER USERS CLUB NL.

# C.U.C. INFO

SVI-COMPUTER GEBRUIKERS MAGAZINE



- MSX
- softwarenieuws
- programma's
- lezerspost
- tricks, tips, hints



- techniek
- markt

MAART 1985

**MSX EXTRA**

POSTBUS 202 - 2300 AE LEIDEN

6

## VOORBEELDEN

---

```
100 REM Vergeten SCREEN 1 te geven
110 ON ERROR GOTO 500
120 CIRCLE (128,96),60
130 GOTO 130
500 REM Foutafhandeling
510 IF ERR=5 AND ERL=120 THEN SCREEN 1: RESUME
520 ON ERROR GOTO Ø
```

---

```
100 REM Leesfout bij inlezen van bestand
110 ON ERROR GOTO 500
120 OPEN "CAS:BESTND" FOR INPUT AS 1
130 FOR T=1 TO 10: INPUT #1, T$(T): NEXT
140 CLOSE 1
150 END
500 IF ERR{<19 OR (ERL{<120 AND ERL{<130) THEN
    ON ERROR GOTO Ø
510 CLS: PRINT "Leesfout, spoel de band terug."
520 RESUME 120
```

---

```
100 REM Kijk of DISK file al bestaat, zo ja,
110 REM vraag of deze overschreven mag worden.
120 ON ERROR GOTO 500
130 CLS: INPUT "Naam van bestand"; N$
140 OPEN "1:" + N$ FOR INPUT AS 1: CLOSE 1
150 CLS: INPUT "Bestaat al! Overschrijven";a$
160 IF A${"J" AND A${"j" GOTO 130
170 OPEN "1:" + N$ FOR OUTPUT AS 1
180 FOR T=1 TO 10: PRINT #1, T$(T): NEXT
190 CLOSE 1
200 END
500 REM Controleer op 'File not found in 140'
510 IF ERL=140 AND ERR=53 THEN RESUME 170
520 ON ERROR GOTO Ø
```

---

-----  
Foutafhandeling  
-----

## ON ERROR GOTO regelnummer

Na deze instructie zal de Basic Interpreter iedere keer dat er een fout in het programma ontstaat naar de aangegeven regel springen.

Daar aangekomen zijn er twee variabelen beschikbaar : ERR (het nummer van de fout) en ERL (het regelnummer waarin de fout zit). Welk nummer bij welke fout hoort, vindt u dan in de (engelse) handleiding. Zo heeft 'Device I/O error' bij voorbeeld foutcode 19.

Nu is er de mogelijkheid om voor diverse fouten zelf een oplossing (eigen foutmelding of een programma om de fout ongedaan te maken) te programmeren. Door ERR en ERL met een bepaalde waarde te vergelijken kunt u uitzoeken welke fout er in welke regel ontstond.

Indien er geen oplossing voorhanden is, moet u 'ON ERROR GOTO 0' geven, waarna het programma met een standaard foutmelding onderbroken wordt.

## RESUME

Nadat u voor een bepaalde fout een oplossing hebt laten uitvoeren, zal het programma weer zijn normale weg moeten gaan. Dit verzorgt de RESUME instructie. Geeft u alleen 'RESUME', dan wordt de instructie waarin de fout optrad opnieuw uitgevoerd, terwijl 'RESUME NEXT' het programma met de eerste instructie na de fout vervolgt. Desgewenst kan ook een regelnummer opgeven ('RESUME regelnummer') worden om aan te geven waar het programma verder moet gaan.

## ERROR

Het is tevens mogelijk ergens in het programma expres een fout op te wekken door een 'ERROR code' te geven. Na ERROR 5 volgt bijvoorbeeld de 'Illegal function call' foutmelding.

C.U.C.-info is het contactor-  
gaan van de SpectraVideo Com-  
puter Users Club NL..

REDACTIE: Wouter Alexander  
Peter Zevenhoven

BasiCode: Rini Kikkert

C.U.C.-consuls:

Oost Ned.: Henk Hoff  
Westeinde 122  
7671 CE VRIEZENVEEN  
tel. 05499 - 2436

AMSTERDAM: Hans v d Meer  
N.W. Ned. Vechtstr 78 hs  
1079 JL AMSTERDAM  
tel. 020 - 445841

DORDRECHT: Peter v Ginneken  
Z.W. Ned. J v Brakelstr 19  
3317 VJ DORDRECHT  
tel. 078 - 183926

Lezerspost, bestellingen, pro-  
gramma's, tapes, documentie,  
copy, manuscripten, adverten-  
ties, lidmaatschap, donateurs,  
en abonnementen:

SpectraVideo C.U.C.  
Postbus 202  
2300 AE LEIDEN

Betaling lidmaatschapsgeld, do-  
naties, advertenties, bestel-  
lingen, documentatie en lezers-  
service artikelen:

Ned. Middenstandsbank  
t.n.v. SpectraVideo C.U.C.  
Rek. nr. 67.86.10.231  
(giro bank 47139)

= Copyrights (85) C.U.C. NL =

ARTIKELEN

- 7 Club reportage
- 10 TEST \*MSX\* adapter
- 13 MSX extra
- 14 MSX software
- 21 SIN & COS
- 25 (V)POKE, (V)PEEK

PRAKTIJK

- 5 PISTOL
- 28 COLLISION

DIVERSEN

- 1 Collofon
- 1 Inhoud
- 3 Redaktorial

RUBRIEKEN

- 4 Lezers in de pen
- 9 BASIC bulletin 7
- 12 DISCOUNT
- 18 BASIC bulletin 8
- 24 AGENDA
- 30 Basicode Corner

EXTRA

BASIC informatie - kaart 5  
MSX pag. 13 - 20

Zonder schriftelijke toestem-  
ming mag niets uit deze uit-  
gave worden overgenomen of  
vermenigvuldigd.

CONSUMENTENLIJN:  
vrijdags 10 - 11 uur  
020 - 139960  
Dhr. d v Roest

# REDAKTORIAL



Geachte lezer,

Aan ons magazine is een nieuwe dimensie toegevoegd! In onze huis-  
kleur, licht blauw, treft U binnen  
in een nieuw gedeelte aan, dat spe-  
ciefiek aandacht aan MSX zal gaan  
weiden. In ons vorige info hebben  
we reeds uitgelegd waarom, en u  
merkt inmiddels, dat juist daardoor  
niemand zich iets te kort gedaan  
zal voelen.

In verband met de PC RAI wordt er  
een extra grote oplage van dit info  
uitgebracht, hetgeen vele nieuwe  
leden zal toevoegen aan ons reeds  
respectabele aantal. Wij beloven de  
eerste mijlpaal in dit opzicht niet  
ongemerkt te laten voorbijgaan.

Maar, waarom streven wij nu eigen-  
lijk naar zo veel leden? Dat is,  
omdat door meer leden meer goed  
bruikbare software van een hoog  
gehalte gaat komen. De listings in  
onze info's, en nu ook onze tweede  
cassette, geven daarin reeds een  
duidelijk inzicht.

Het parool blijft daarom: zend aan  
je C.U.C. redactie toe wat je hebt,  
want velen zullen 'er dan met ple-  
zier van kunnen profiteren.

*Wouter Alexander*

Hello,  
I'm interested in your computer club and wish to be a member of it.  
Please, send me details as soon as possible.

Asko Lippo  
FINLAND.

Red.: Hier zijn wij erg blij mee; uw C.U.C. begint internationaal bekend te worden.

\*\*\*

Beste C.U.C.,  
hier volgen enkele listings naar mijn idee aardigheidjes. Dit naar aanleiding van het uitbrengen van uw fantastische clubblad, dat altijd om aardigheidjes vraagt.

Hans Klopper  
Amsterdam

Red.: Bedankt, Hans. We kunnen ze niet allemaal tegelijk plaatsen, maar je hebt goed begrepen wat wij bedoelen. Wie volgt ?!

\*\*\*

Mijne heren,  
Het gebruikers magazine is ook in Zuid-Limburg goed ontvangen. Ik, en vele Spectra-video gebruikers met mij, ben zeer enthousiast over uw blad en zie vol verwachting uit naar de volgende nummers. Bijgaand zend ik u een tweetal programma's.  
Groetend met een zachte G,

Ron Wehrens  
Sittard.

Beste SV users club,  
Na het lezen van uw laatste magazine besloot ik ook eens een programmaatje in te sturen.

Ik kwam nl. tussen de "aardigheidjes" een programmaatje tegen dat een spiraal tekent en erg eenvoudig gemaakt was. Na wat diep denkwerk kwam ik tot een programmaatje dat een logaritmische spiraal tekent.

Om maar meteen twee programma's in te sturen (een is zo weinig, drie is te veel tikwerk) heb ik het volgende uitgezond. Ik hoop ook, dat jullie een extra dik juli nummer kunnen maken. Veel succes!

Frans Hondeman  
Amsterdam

Red.: Kennelijk doen de aardigheidjes het goed en ook deze reactie stellen wij zeer op prijs!


Iedereen zal natuurlijk wel begrijpen, dat het slagen van het grote en dikke juli nr. niet alleen van de redactie afhankelijk is maar ook van de (programmeer) werkzaamheid van de club magazine lezers.

\*\*\*

Intussen heeft ook een in 't frans geschreven zusje van het C.U.C.-info het licht gezien. In de toekomst zullen we daar nog nader op terugkomen.

Sturen jullie nog wat aardigheidjes??

Red.



```

1000 REM Pistool
1010 :
1020 REM Dit programma zoekt zelf uit of het in een SVI.318/.328
1030 REM dan wel een SVI.728 zit en handelt daarnaar.
1040 :
1050 REM MSX=0: SVI.318/.328, MSX=1: SVI.728 of andere MSX comp.
1060 MSX=0: IF PEEK(0)=6FH3 THEN MSX=1: OPEN "grp:" AS 1
1070 VL$=CHR$(201-MSX): ' Grafisch teken, vlakje van 6*8 dots
1080 :
1090 REM Initialisatie en introductie plaatje
1100 SOUND 7,30: PLAY "t200 L4 v10 o4 codecc o3 ab n36 ffa b1
1110 ON STOP GOSUB 2290: STOP ON
1120 COLOR 15, 1, 1: SCREEN 1+MSX
1130 REM Buitenste C
1140 CIRCLE (112, 60), 50, , .65, 5.66
1150 CIRCLE (112, 60), 42, , .78, 5.55
1160 LINE STEP (32,-27) - STEP ( 9, 0)
1170 LINE STEP ( 0, 54) - STEP ( -9, 0)
1180 PAINT STEP ( 0, 2): COLOR 7
1190 REM Middelste U
1200 CIRCLE (112, 60), 35, , 2.58, .59
1210 CIRCLE (112, 60), 27, , 2.45, .74
1220 LINE STEP ( 30, -18) - STEP ( -9, 0)
1230 LINE STEP (-43,  0) - STEP ( -9, 0)
1240 PAINT STEP ( 1, 2): COLOR 3
1250 REM Binnenste C
1260 CIRCLE (112, 60), 20, , .55, 5.8
1270 CIRCLE (112, 60), 13, , .81, 5.51
1280 LINE STEP ( 9, -9) - STEP ( 8, 0)
1290 LINE STEP ( 1, 17) - STEP ( -9, 0)
1300 PAINT STEP ( 2, 2): COLOR 15
1310 REM Tekst
1320 FOR X=0 TO 1: HO=160+X: VE=35
1330 PR$="H. P. Zeedijk": GOSUB 50000
1340 NEXT
1350 HO=189: VE=51: PR$="C.U.C.": GOSUB 50000
1360 HO=184: VE=65: PR$="VAASSEN": GOSUB 50000
1370 HO=60: VE=120: PR$="P R E S E N T E E R T ": GOSUB 50000
1380 COLOR 13: HO=70: VE=140: PR$="PISTOOL-SCHIETEN."
1390 GOSUB 50000: COLOR 3
1400 HO=65: VE=160: PR$="Voor stoppen druk ESC.": GOSUB 50000
1410 HO=40: VE=175: PR$="WILT U BEGINNEN? DRUK VUUR-KNOP!"
1420 COLOR 13: GOSUB 50000
1430 REM Wacht op indrukken van vuurknop
1440 IF NOT (STRIG(0) OR STRIG(1)) GOTO 1440
1450 :
1460 REM Beeldscherm opbouw
1470 COLOR 15, 1, 6: CLS: PR$="HI-SCORE: "+STR$(HIS)
1480 HO=93: VE=35: GOSUB 50000: COLOR 10
1490 LINE(80,7)-STEP(108,11),4,BF
1500 HO=84: VE=9: PR$="Pistool-schieten.": GOSUB 50000
1510 REM Pistool tekenen
1520 DRAW "bm10,100 nr2 d3 r1 nu3 r1 nu3 f3 e6 r24 d3 r30 u3
1530 DRAW "r6 d8 L39 d3 ng2 u3 L9 d3 nr5 d2 nr6 d3 L3 nd5 L3
1540 DRAW "d5 nr3 d4 g8 L12 u8 e8 u10
1550 PAINT ( 20, 105)
1560 GOTO 1730

```

```

1570 :
1580 REM Kogelbaan
1590 LINE (256, 105) - (100, 105), 15
1600 IF Y<=104 AND Y=98 GOTO 1910
1610 :
1620 REM Gemist
1630 FOR PAUZE=1 TO 5: NEXT PAUZE
1640 LINE (256, 105) - (100, 105), 1
1650 MIS=MIS+1
1660 LINE (10, 74) - STEP( 85, 9), 4, BF: COLOR 14
1670 HO=10: VE=75: PR$=STR$(MIS)+" x gemist": GOSUB 50000
1680 IF MIS = 30 THEN GOTO 2120
1690 FOR I=0 TO 500: NEXT I
1700 COLOR 1: HO=X: VE=Y: PR$=VL$: GOSUB 50000
1710 :
1720 REM Doelen: positie bepalen en plaatsen
1730 A=INT(RND(1)*4+1)
1740 ON A GOTO 1750, 1760, 1770, 1780
1750 X=250: FOR Y= 0 TO 196 STEP 4: GOTO 1800
1760 X=250: FOR Y=196 TO 0 STEP -4: GOTO 1800
1770 X=256: FOR Y= 0 TO 196 STEP 4: X=X-4: GOTO 1800
1780 X= 58: FOR Y= 20 TO 196 STEP 4: X=X+4: GOTO 1800
1790 REM Plaats doel
1800 COLOR 10: HO=X: VE=Y: PR$=CHR$(193+56*MSX): GOSUB 50000
1810 REM Geschoten? Zoja, kogelbaan
1820 IF STRIG(0) OR STRIG(1) THEN GOTO1590
1830 REM <ESC> gedrukt? Zoja, stoppen
1840 IF INKEY$=CHR$(27) GOTO 2120
1850 REM Wis doel
1860 COLOR 1: GOSUB 50000
1870 NEXT Y
1880 GOTO 1720
1890 :
1900 REM Raak geschoten
1910 COLOR 1: HO=X: VE=Y: PR$=VL$: GOSUB 50000
1920 COLOR 15: PR$="*": GOSUB 50000
1930 REM Knal
1940 SOUND 0, 0: SOUND 1, 5: SOUND 5, 15: SOUND 3, 13
1950 SOUND 4, 255: SOUND 5, 15: SOUND 6, 30: SOUND 7, 0
1960 SOUND 8, 16: SOUND 9, 16: SOUND 10, 16: SOUND 11, 0
1970 SOUND 12, 5: SOUND 13, 0
1980 FOR I=1 TO 30: NEXT I
1990 SOUND 12, 56: SOUND 13, 0
2000 FOR I=0 TO 200: NEXT I
2010 LINE(120, 85) - STEP (60, 8), 1, BF
2020 RAAK= RAAK+1
2030 LINE ( 10, 134) - STEP ( 85, 9), 4, BF
2040 COLOR 14: PR$=STR$(RAAK)+" x geraakt"
2050 HO=10: VE=135: GOSUB 50000
2060 FOR I=200 TO 500: NEXT I
2070 LINE (100, 105) - (256, 105), 1
2080 COLOR 1: HO=X: VE=Y: PR$=VL$: GOSUB 50000
2090 GOTO 1730
2100 :
2110 REM Op <ESC> gedrukt: score bepalen
2120 IF MIS*RAAK THEN SC=INT(100/MIS*RAAK) ELSE SC=0
2130 REM Topscore?
2140 IF SC>HIS THEN HIS=SC
2150 LINE (93, 35) - STEP (90, 8), 1, BF: COLOR 15
2160 HO=93: VE=35: PR$="HI-SCORE: "+STR$(HIS): GOSUB 50000
2170 COLOR 1: HO=X: VE=Y: PR$=VL$: GOSUB 50000

```



```

2180 REM Print score
2190 LINE (95, 101) - (204, 114), 3, B: COLOR 15
2200 HO=100: VE=105: PR$="Uw score is:"+STR$(SC): GOSUB 50000
2210 :
2220 REM Nog eens?
2230 HO=10: VE=100: PR$="Wilt u nogmaals (j/n)?:": GOSUB 50000
2240 Z$=INKEY$: IF Z$="n" OR Z$="N" THEN 2290
2250 IF Z$="j" OR Z$="J" THEN SC=0: RAAK=0: MIS=0: GOTO 1470
2260 GOTO 2240
2270 :
2280 REM Na CTRL/STOP of 'N':
2290 COLOR 15,4,5
2300 END
2310 :
50000 REM Subroutine om PR$ op positie HO en VE te printen
50010 REM met een onderlinge afstand van 6 dots per taken
50020 REM (dat is de standaard SVI.318/.328 afstand).
50030 :
50040 IF MSX=0 THEN LOCATE HO,VE: PRINT PR$: RETURN: ' 318/328
50050 :
50060 REM Voor MSX computers:
50070 FOR Q9=1 TO LEN(PR$)
50080 Q8=6*Q9-6+HO: DRAW "bm=Q8;, =VE;"
50090 PRINT #1, MID$(PR$,Q9,1)
50100 NEXT
50110 RETURN
50120 :
50130 ' Originele programma: H. P. Zeedijk, VAASSEN
50140 ' Aanpassing aan MSX (gosub 50000 enz), extra REM
50150 ' instructies en spaties: P.Zevenhoven

```



==CLUB-REPORTAGE==

RED.

Om maar gelijk met de bit in de byte te vallen..... we hebben kunnen vaststellen dat er onder onze leden inmiddels een behoorlijk aantal dames lid zijn, en dat de belangstelling van hen voor de clubdagen behoorlijk is toegenomen. We zijn daarmee zeer vereerd, en zouden het nog meer toejuichen indien er ook uit deze groep artikelen c.q. computerprogramma's komen; het lijkt niet uitgesloten dat een vrouw de computerwereld toch anders bekijkt. U bent van harte uitgenodigd, dames!

Nu het punt computerprogramma's toch genoemd is even het volgende. Het aantal ingezonden listings naar Postbus 202 te Leiden is de afgelopen maand aanzienlijk toegenomen en het niveau is stijgend. Maar, we kunnen niet alles direkt plaatsen. Het is zelfs nodig te selecteren en enkele programma's noodzakelijkerwijs aan te passen. Zoals u wellicht uit de reeds gepubliceerde listings gewerkt zult hebben, bestaat er

een soort "huisstijl". Indien u software wilt insturen graag bij voorkeur op een cassette (niet perse noodzakelijk), maar wel:

- a) voorzien van duidelijke tekst in het programma (REM): we hebben (vinden wij) ook een educatieve taak;
- b) regelnummers gaarne beginnen vanaf 100 (basicode 1000), en per 10 olopend;
- c) aan het eind uw naam, adres en eventueel telefoonnummer vermelden;
- d) het belangrijkste: maak de regels niet te lang: mocht er een of andere typefout in staan, dan zoekt men zich blauw (afhankelijk van kleur der monitor.....bij de samensteller van dit artikel amber).

Nu de letters MSX voor het publiek een begrip beginnen te worden (voor ons was dat reeds lang het geval), komen er steeds meer boeken en software (computerprogramma's) in de handel.

De laatste tijd horen wij hierover ook minder gunstige berichten, maar de goede boeken en programma's zullen wij gaarne vermelden/bespreken. We wijzen erop, dat tussen deze artikelen ook lektuur voorkomt die wij het predicaat "slecht/zeer slecht" zouden willen geven. We zullen deze nog niet in ons orgaan vermelden, want smaken verschillen en wij zijn zeer kritisch - ook voor onszelf.

Derhalve bij deze een wijze raad: koop alleen boeken of software welke u hebt gezien of wordt aangeraden door deskundigen.

Onlangs ontvingen we van de fa. ERA te Leiden vrijblijvend een ruime sortering Academic Service computerboeken. We dachten, dat een aantal zeer interessant is voor u en bestuderen deze voor toekomstige recenties.


Tenslotte een veel gehoorde klacht:"wat was het clubblad laat!!!" of "bestaan jullie nog wel?"..... terecht, maar vergeet u niet, dat het blad door slechts 3 mensen wordt samengesteld!!!!

- 8 We zijn echter "achter de schermen" heel hard bezig dit euvel te verhelpen, en hebben daarom ook uw hulp HEEL hard nodig. Over de "grootse" plannen gaan we u op dit moment nog niets verklappen, doch uw vertrouwen in de club zullen we zeker niet beschamen.

STUUR WEL Z.S.M. EEN ARTIKEL IN OMTRENT UW ERVARING MET BIJV. DISKS, PROGRAMMA'S, E.D. AAN DE REDAKTIE.

Dit is een one liner die regels uit de driehoek van Pascal oproept.

```
1 WIDTH40:CLS:DIMA(255),B(255):INPUT"Ri
jnr. ";Y:A(O)=1:A(1)=1:B(O)=1:FORX=1TOY:
B(X)=A(X-1)+A(X):IFB(X)<>1THENNEXTELSEB
(X)=1:FORA=OTOY:A(A)=B(A):NEXT:K=K+1:IF
K<Y-1THENX=0:NEXTELSEFORX=OTOY:PRINTUSI
NG"*****";A(X):NEXT:PRINT
```



```
100 REM vierkanten - WA _
110 COLOR,14,11: SCREEN 1
120 LINE(5,0)-(255,192),1,B
130 X=50:Y=175:A=75:B=150
140 M=210:N=15:K=185:L=40
150 X=X+2:Y=Y-2:A=A+2:B=B-2
160 M=M-2:N=N+2:K=K-2:L=L+2
170 LINE(X,Y)-(A,B),4,B
180 LINE(M,N)-(K,L),8,B
190 IF X>110 THEN 200 ELSE 150
200 X=115:Y=80:A=140:B=105
210 M=116:N=81:K=141:L=106
220 X=X-2:Y=Y-2:A=A-2:B=B-2
230 M=M+2:N=N+2:K=K+2:L=L+2
240 LINE(X,Y)-(A,B),1,B
250 LINE(M,N)-(K,L),13,B
260 IF X<50 THEN 270 ELSE 220
270 FOR TI= 1 TO 2000: NEXT TI
280 LINE(7,3)-(250,189),14,BF
290 GOTO 130
300 END
```

```
100 REM -grafieken tekenen-
110 COLOR,12,11: SCREEN 1
120 LINE(5,0)-(256,192),12,BF
130 FOR Y=0 TO 197 STEP 49
140 LINE(6,Y)-(253,Y),1
150 NEXT Y
160 FOR X= 5 TO 256 STEP 50
170 LINE(X,0)-(X,190),1
180 NEXT X
190 LOCATE 60,15: PRINT"grafiek"
200 FOR A=-15 TO 15 STEP .1
210 PSET(A+ 38,A*A+5*A+45),15
220 PSET(A+130,A*A+5*A+90),15
230 PSET(A+235,A*A+3*A+10),15
240 NEXT A
250 FOR T=1 TO 300: NEXT T
260 GOTO 120
270 REM = W&T T. - W.A. =
280 END
```

Voor MSX geldt: SCREEN 1 of 2 wordt SCREEN 2 of 3

Natuurlijk daarna RUN en ENTER geven.

samenstelling: Wouter Alexander

# MSX GAME ADAPTER

artikel : Wouter Alexander  
techniek : Peter Zevenhoven  
praktijk : Rini Kikkert

## De adapter

Lang verwacht en eindelijk gekomen: de Spectravideo MSX adapter. Dit stuk electronisch vernuft maakt het mogelijk MSX spellen op de .318(II)/328(II) te draaien en ook wat typische MSX hardware aan te sluiten. We hebben de adapter aan de tand mogen voelen waarvan onderstaande impressie het resultaat is.

Wat uiterlijk het eerste opvalt, is het grote cartridge slot bovenop. Minder opvallend zijn de twee rechts in de zijkant aangebrachte joystick aansluitingen en achterin de DIN aansluiting voor een normale handels cassette recorder. Aansluitingen voor een disk drive en een printer ontbreken, hetgeen zijn oorzaak vindt in de hardware opbouw van MSX.

Wat innerlijk opvalt, is de enorme hoeveelheid complexe electronica: 32K MSX ROM + 8K aanpassings ROM, 16K RAM, de geluidsgenerator, 23 TTL IC's en een 8255 PPI. Zowaar een wereldje op zichzelf.

## De recorder aansluiting

Het bleek niet mogelijk direct met een standaard DIN kabel aan de gang te kunnen gaan. De 8-pins DIN plug op de adapter is nogal afwijkend aangesloten. Pin 1,2 en 3 liggen aan aarde, 4 en 5 zijn voor LOADen en SAVEn; de rest voor motor control. In feite zou de DIN aansluiting op de recorder ook veranderd moeten worden, want voor LOADen moet men het geluid aftakken van de luidspreker (of ear).

Om met een gewone recorder te kunnen werken hebben we dus zelf een kabeltje gemaakt, maar het behaalde resultaat was niet zo bevredigend, omdat het LOADen en SAVEn nogal erg kritisch luistert. We raden daarom ook aan met bijv. de SVI.767 MSX DATA RECORDER te werken of met een recorder die voorzien is van een aparte aansluiting voor MIC, EAR en REMOTE (en liefst ook nog een tellertje). Spectravideo biedt een speciaal kabeltje hiervoor aan (SVI.208).

Tijdens het LOADen en SAVEn wordt men onmiddellijk attent gemaakt op het klikken van het recorder aan/uit relais in de adapter: de SOUND ON/OFF instructie kent MSX niet, wat voor degenen die 't gewend zijn wel even wennen is.

### Printer en disk drive

Het aansluiten van een printer (of extra RAM) via het slot is niet mogelijk. Met een aangesloten (Sony) disk drive konden we wel LOADen en SAVEn, RUNnen was er niet bij. Idem dito gold dit voor de nieuwe Quick Drive. Jammer, maar een feit.

### MSX BASIC

Bij het opstarten van de .328(II) met aangesloten adapter verschijnt niet het (voor ons zo vertrouwde) Spectravideo logo op het scherm. Uit alles blijkt inderdaad direct, dat men onder MSX werkt. De reden hiervoor ligt opgesloten in het in de adapter ingebouwde MSX ROM van de .728, dat totaal afwijkt van het SVI BASIC ROM. Daarnaast is er nog de 8K aanpassings ROM die de computer iets (ca. 15%) trager maakt. Wanneer MSX zich meldt, maken we grote ogen als het scherm opgeeft dat er 12431 Bytes free zijn. Niet slechts het ROM van de .318/.328(II) worden uitgeschakeld, ook het RAM is niet meer aanspreekbaar en we moeten volledig steunen op de 16K RAM in de adapter waarvan er 12K bruikbaar overblijven.

De graphics zijn geheel volgens de MSX standaard (hoewel iedere fabrikant daar aan mag toevoegen).

### De geluidsgenerator

Opvallend was het zeer lage niveau van de ingebouwde geluidsgenerator. Om PLAY instructies voldoende luid te kunnen weergeven, moet de volume regelaar zo ver worden openge draaid, dat bij de eerste de beste "toetsenbord klik" ons hart een slagje oversloeg.

### Algemeen

Een MSX programma dat zich niet geheel aan de MSX afspraken houdt (dat bijv. zelf het toetsenbord decodeert i.p.v. het ROM BIOS te gebruiken), of dat er van uit gaat meer dan 16K RAM in de computer aan te treffen, zal niet goed of helemaal niet kunnen draaien via de adapter. Een zaak om rekening mee te houden.

### Conclusie

Wat er in de adapter opgesloten zit, is niet niks. Op het key board (in België klavier [of clavier]), de videoprocessor en cpu na, is het een complete computer die met behulp van de 8K aanpassings ROM toch wel knap gemaakt is. Maar voor de (aangepaste) adviesprijs van f 249,- krijgt men natuurlijk wel iets in huis!! Leverbaar kort na de PC RAI.

Spectravideo draagt het zelf al aan, het gaat om een game adapter en er zijn dan ook duidelijke wat beperkingen. Max. 16K gebruikers RAM (al is dat nog dubbel de MSX norm eis) is aan de magere kant en ook het boven genoemde toetsenbord scan probleem kan een handicap vormen. Vooraf uitproberen van de software bij aanschaf is zeer raadzaam in dit geval. Afgezien hiervan draaien BASIC en machinetaal programma's via de adapter uitstekend.

////////////////////////////////////  
\*\*\* DISCOUNT \*\*\*

Dit wordt een nieuwe rubriek waarin u uw hardware te koop aan kunt bieden. Voor f 10,- plus uw volledige advertentie bereikt u een groot kooppubliek

SV.318/.28 bezitters zonder printer: stuur programma op cassette naar onderstaand adres. Voor enkele centen per regel ontvangt u enkele dagen later uw listing en/of tekening.  
I. Buné Buizerdhorst 58  
5431 LX CUIJK.

Wegens overgang op bedrijfs PC te koop: SV328, Philips V7001(gr), expand. SV.601, disk contr. SV.801, Centron int. SV.802, drive SV.902, rec. SV.903, print. SV3000, handleid. en 3 extra boeken in één koop f 2900,--.  
Tel. 01864 - 2130.

```
100 REM Patroon
110 IF PEEK(0)=195 THEN SCREEN 1
    ELSE SCREEN 2
120 FOR X=0 TO 255 STEP 2
130 A=A+2
140 LINE(X, A)-(255-X, 192-A), 15, B
150 NEXT X
160 GOTO 160
170 END
```



\*\*\*\*\* Stuur ons uw software! \*\*\*\*\*

SPECTRAVIDEO COMPUTER USERS CLUB NL.



**MSX EXTRA**

SVI MSX COMPUTER WETENSWAARDIGHEDEN



Van veel computer enthousiasten is de belangstelling uitgegaan naar de MSX types. Een groot deel van hen koos voor de Spectravideo SV.728 MSX computer, en niet ten onrechte. Vervolgens, echter, komt de "nazorg", de ruggesteun die een computer gebruiker moet kunnen vinden om zijn hobby zo doeltreffend mogelijk machtig te kunnen worden. Dat nu is en zaak gebleken voor de computer clubs.

Alleen, op het MSX terrein zijn er zo veel, inmiddels. Een MSX club ziet meestal leden met computers van allerlei pluimage, zoals bijv. ook een CP/M g.g. dat kent. In het laatste geval vindt ik dat niet zo problematisch, maar het eerste geval eist toch wel wat meer aandacht, vooral voor Spectravideo MSX hobby computeraars. Immers, het Spectravideo (SV) BASIC was de bron waaruit MSX werd geboren en is reeds enkele jaren o.a. bij het C.U.C. vertrouwd en gekend. Die ervaring is exclusief en duidelijk een aanbeveling voor deze computer club.

Wat CP/M betreft, SV computers draaien dit wereldbekende besturingssysteem en we maken er in deze uitgave een begin mee het wat nader uit de doeken te doen. Uiteraard is dit ook interessant voor alleen CP/M machines, zoals de Bondwells. Op deze wijze is ook de mogelijkheid aanwezig om samen aan de opbouw van een eigen CP/M programma bibliotheek te werken.

**C.U.C.**

**7**

POSTBUS 202 - 2300 AE LEIDEN

## SOFT/HARDWARE voor MSX

De nationale softwarehuizen zijn inmiddels volop bezig op de Nederlandse MSX markt. Naast de nodige spelletjes op cassette, of in rom-pack, is hiervoor zelfs nu al speciale software, zoals bijv. een cassette met een belastingprogramma voor de aangifte van uw Inkomsten belasting 1984 (uitstel aangevraagd?!) verschenen. Wij ontvingen van de uitgeverij Kluwer te Deventer een exemplaar en hebben deze in een SV.728 geladen om het te bekijken (helaas konden wij hierover net niet meer in ons vorige nummer schrijven).

Het programma is ongeveer 26K lang (niet gering) en visueel eenvoudig gehouden: alles werkt op een zwarte achtergrond met (kleine) groene letters. De indeling van de tekst op het scherm is overzichtelijk en men verwijst regelmatig naar de punten op het aanslagbiljet.

Nadat alles is ingevuld, volgen er enkele tussen overzichten die de totalen geven voor de gehele aangifte. En tenslotte rekent het programma uit wat de teruggave wordt of moet worden bijbetaald.

Het is wel even wennen met het invullen van de cijfers, maar vooraf het programma doorlopen met gefingeerde cijfers maakt alles duidelijk. De gegevens kunnen op een printer gezet worden, dus zeer eenvoudig om later even terug te slaan. Hoewel het programma zonder Kluwers gids kan werken, wordt daar soms wel naar verwezen voor een nadere toelichting in een bijzonder geval.

Mocht u uw aangifte nog niet hebben ingevuld (uitstel aangevraagd), dan kunt u de cassette in de boekhandel e.d. aanschaffen voor f 29,50.

### SV-606 MSX GAME ADAPTER

In ons vorige info hebben wij een test geplaatst van deze adapter.

We hebben inmiddels ook proeven gedaan met de SV-1400 datarecorder (verderop hierover meer) i.p.v. een gewone recorder met DIN-aansluiting en deze zijn zeer positief verlopen.

Helaas blijkt het aantal rom-packs dat niet werkt met de adapter aanzienlijk te zijn. Veel "Konami"-spelen bijv. doen het niet; vooraf uitproberen bij de dealer is raadzaam. Schuld: MSX norm wordt softwarematig niet aangehouden.

### DATARECORDER SV-1400

Om programma's van de SV.728 (en andere MSX-computers) te kunnen bewaren, heeft men een recorder nodig. De sinds enige tijd op de markt aanwezige SV-1400 is er een van een nieuwe generatie betrouwbare datarecorders. Deze bezit zelfs een speciale multi-standaard constructie, zodat ook veel andere merken computers met de recorder kunnen werken.

Opvallend zijn de speciale LEDs: met de schuifbare volumeregeling kan men zichtbaar het beste resultaat verkrijgen. Handig is verder het ESS (Electronic Search System) waarmee men snel naar programma's kan zoeken, daar het een ingebouwde monitor bezit, zodat men zowel bij het SAVEN als LOADEN mee kan luisteren.

De recorder heeft (helaas) geen lichtnet aansluiting; wel werkt hij op 4 kleine batterijen en via een lichtnet adapter (in te stellen op 6 volt). Dit laatste raden wij aan. Het LOADEN en SAVEN gaf totaal geen problemen. We adviseren u niet te lange cassettes te gebruiken en zeker geen chroom, ferrochroom of metal bandjes. Dit geldt trouwens voor alle opnamen van computerprogramma's op cassettes.

Bekijk deze recorder eens bij uw dealer en u zult er spoedig "verliefd" op zijn.



Wat weten we in de eerste plaats van cirkels? Met de opdracht CIRCLE zijn ze op SCREEN 1 en 2 te plaatsen. Weliswaar meer ovaalachtig, maar met een correctiefactor hebben we de beschikking over mooie, ronde cirkels

```
100 SCREEN 1
110 CIRCLE(128,96),50,1,,1.3
120 GOTO 120
130 END
```

wat precies onze bedoeling was (att.: voor MSX SCREEN 2 of 3 kiezen).

Waar we even bij stil kunnen staan, is, dat vroegere computers deze functie niet kenden. Zij moesten met veel moeite via veel gereken in een subroutine op de e.o.a. manier maar zien een cirkel op het scherm te voorschijn te roepen.

Weliswaar moet dat gereken toch wel gebeuren, echter, het gebeurt heel onopvallend, omdat de computer dit werk automatisch voor zijn rekening neemt. Immers, na de opdracht CIRCLE tekent de computer niet slechts, veel meer gaat hij onmiddellijk aan het rekenen. Hij zal in dit geval herhaaldelijk de SIN en COS functies moeten gebruiken en dat neemt, vooral bij dubbel precisie getallen, nogal wat tijd. Vandaar, dat het tekenen van een cirkel in BASIC toch nog wel even wat tijd vergt.

Omdat het BASIC van de Spectravideo computers de goniometrische functies SIN (sinus) en COS (cosinus) - en nog veel meer - ondersteunt, kunnen we zelf ook wel cirkels op de SCREENs componeren en daarna besluiten dat we toch maar gelukkig af zijn met ons SVI super extended BASIC. Alles gaat weliswaar nog wat trager dan de CIRCLE opdracht, maar educatief is het wel.

Op de eerste plaats dienen we dan te beschouwen wat de SIN en COS functies zijn en doen. Wat ze "zijn" is een verhouding uitbeelden, nl. die van de zijden van een rechthoek (zie fig. 1). SIN is de verhouding van zijde A en B (A/B), en COS van de zijden C en B (C/B). Indien we het getal dat zo'n verhouding oplevert vermenigvuldigen met een variabele

(in fig. 1 bijv. R, de diameter van de cirkel) dan verandert die variabele, hij krijgt een nieuwe waarde. En met die waarden kunnen we aan ons doel gaan werken.

Wees niet benauwd dat we hier goniometrisch wiskundige acrobatiek gaan uithalen. Waar het om gaat, is waarden te verkrijgen voor de PSET instructie. Alleen zodanig, dat de opeenvolgende PSET posities op het scherm niet een lijn, maar een cirkelvorm gaan aannemen.

Het betreft hier de X- en Y- coördinaten, dus PSET(X,Y). Ieder punt op een cirkel wordt bepaald door twee goniometrische functies:

X-pos.= straal cirkel x sin(hoek)  
en Y-pos.= straal cirkel x cos(hoek).

Derhalve, wanneer we in BASIC met PSET een cirkel willen opbouwen, zullen we de waarden van de X- en Y- coördinaten van ieder punt op de cirkelomtrek met behulp van SIN, COS en de lengte van de straal vast moeten stellen. Dat is wat deze functies "doen".

Wat bijv. de SIN doet blijkt wel uit:

```
100 COLOR,6,1: SCREEN 1
110 ST=30: 'ST=straal cirkel
120 FOR HOEK= 1 TO 360 :' 360 graden
130 PSET(128+ST*SIN(HOEK/57.2),HOEK/2),15
140 NEXT HOEK
150 GOTO 150
160 END
```

Voor de uitwerking van COS zouden we in kunnen lassen:

```
135 PSET(35+HOEK/2,122+ST*COS(HOEK/57.2)),15
```

Er komt hier naar voren, dat de straal (R) wordt verkort naar gelang de waarde die de SIN bij die hoek (a) aanneemt. Voor iedere hoekgrootte vinden we een nieuwe waarde, vandaar de golfbeweging die de SIN functie oproept.

Nu gaat het er alleen nog om de curve van de SIN zo te verbuigen, dat er een cirkel ontstaat. Daar SIN de X-pos. berekent, zullen we de Y-pos. nog moeten berekenen en die arbeid verricht COS voor ons. Vandaar, dat een punt op een cirkelomtrek wordt bepaald door deze twee goniometrische functies.

In een FOR-NEXT lus die alle 360 graden van een cirkel doorloopt, gaan we in een programma alle (360) punten op de cirkelomtrek berekenen met de SIN en COS functies:

```
100 COLOR,6,1: SCREEN 1: STRAAL=30
110 FOR HOEK= 1 TO 360
120 X=128+STRAAL*SIN(HOEK)
130 Y= 96+STRAAL*COS(HOEK)
140 PSET(X,Y),15
150 NEXT HOEK
180 GOTO 180
190 END
```

(De waarden 128 en 96 plaatsen de cirkel op 't midden van 't scherm).

Zo helpt dan die moeilijke wiskunde ons meer uit de computer te halen.

Aan een aspect moeten we nog nader onze aandacht schenken.

meestal wordt aangegeven. Het bovenstaande programma tekent de cirkel dan ook op een enigszins vreemde wijze. Om nu simpelweg de cirkel als een ronde lijn, punt voor punt, op te bouwen, veranderen we in het bovenstaande programma SIN(HOEK) in SIN(HOEK/57.2) en COS(HOEK) in COS(HOEK/57.2). Ziet u maar eens voor u zelf nu.

Om te controleren of onze cirkel ook echt dicht is, kunnen we de volgende regels inlassen:

```
160 CIRCLE(128,96),50,1,,,1.3
170 PAINT(128,96),15
```

De alternatieve cirkel blijkt nu prima in orde, maar niet rond te zijn. Aan SIN(HOEK/57.2) toevoegen \*.75, en dat is weer dat.

Tot slot: het samenspel van SIN en COS met als resultaat een cirkel wordt als volgt aanschouwelijk op het scherm gebracht:

```

100 COLOR,6,1: SCREEN 1
110 ST=30: 'ST=straal: HO=hoek
120 LINE(4,0)-(254,191),11,B
130 LINE(10,180)-(245,190),12,BF
140 LINE(10,180)-(245,190),1,B
150 :
160 LOCATE 20,183: PRINT
'SIN & COS bouwen samen een cirkel op"
170 LOCATE 85,15: PRINT "SINUS"
180 LOCATE 15,140: PRINT "COSINUS"
190 LOCATE 145,130: PRINT "cirkel"
200 :
210 FOR HO=1 TO 360 STEP 2
220 X=(128+ST*SIN(HO/57.2))*75)
230 PSET(X,(HO/2)),15 : 'sin
240 Y=98+ST*COS(HO/57.2)
250 PSET(35+HO/2,Y),15 : 'cos
260 PSET(X,Y),15 : 'cirkel
270 NEXT HOEK
280 GOTO 280
290 END

```




FIG. 1

• • • • —

Het tijdschrift Psychology Today vat 25 jaar computer research als volgt samen: "Elke menselijke peuter kan drie dingen doen die nog geen enkele computer presteert - een gezicht herkennen, een natuurlijke taal begrijpen en op twee benen lopen." Laat staan een knipoog in de gaten hebben en snappen. WA

~~~~~**AGENDA**~~~~~C.U.C. gebruikersdagen voor leden:  
 Wij zijn te gast bij:

13 april a.s. bij TEFO - Lubbers te Den Haag

4 mei a.s. bij C.D. Systems te Lisse

U kunt terecht vanaf 11.00 uur + koffie, tot 15.00 uur. Neem al Uw software en listings mee om te ruilen of te kopiëren. Begrijpelijk(er)wijs zal er geen software uit de reguliere handel mogen worden gecopieerd!!

Wij zouden alle clubleden willen uitnodigen hun programma's, ideeën, tips en tricks ter publicatie aan de redactie toe te sturen.

Red.

In de vorige uitgave is de werking van SCREEN 0 besproken. Nu volgt de beschrijving van SCREEN 1 en 2, zoals deze door het systeemprogramma van de SVI. computers gebruikt worden. Voor de SVI.728 moet voor SCREEN 1 en 2 respectievelijk 2 en 3 gelezen worden.

SCREEN 1: het grafische scherm met hoge resolutie

Scherf 1 verdeelt het beeldscherm in 24 regels van 32 blokjes per regel; ieder blokje is acht bij acht pixels groot. Om het aanzien van de blokjes te bepalen zijn er 2 tabellen in VRAM aanwezig: de bitpatroontabel (adressen 0 tot en met 6143) en de kleurentabel (adressen 8192 tot en met 14335).

Iedere acht pixels (van links naar rechts gezien) op het scherm worden bepaald door 1 byte uit de bitpatroontabel (8 nullen of enen) en 1 byte uit de kleurentabel (de linker 4 bits bepalen de kleur als een bit uit het bitpatroon 1 is en de rechter 4 bits bepalen de kleur als dat bit 0 is). Merk op dat er maar 2 kleuren bij deze acht pixels gebruikt kunnen worden.

De eerste 256 bytes van de bitpatroon- (0-255) en kleurentabel (8192-8447) bepalen het aanzien van de bovenste regel, de 256 bytes daarna zijn voor de 2e regel, enzovoorts.

Het volgende voorbeeldprogramma laat deze opbouw zien:

```
100 SCREEN 1
110 FOR T = 0 TO 6143
120 VPOKE T, &B11010000 : ' BITPATROON
130 VPOKE T + 8192, &HB1 : ' KLEUR (geel/zwart)
140 FOR I = 1 TO 10: NEXT : ' VERTRAGING
150 NEXT
160 END
```

SCREEN 2: het grafische scherm met lage resolutie

Scherf 2 heeft 6 regels van 32 blokjes per regel. Ieder blokje is 2 pixels breed en 8 pixels hoog (deze pixels zijn vier keer zo groot als die van scherm 1). Op dit scherm wordt eigenlijk alleen van een kleurentabel gebruik gemaakt

(VRAM adressen 0 tot en met 1535). Een kleurenbyte bepaalt de kleur van 2 pixels op het scherm (de linker 4 bits voor het linker pixel). Adressen 0 tot en met 7 bepalen de kleuren van de 16 pixels linksboven (2 naast en 8 onder elkaar), adressen 8 tot en met 15 doen dat voor de rij ernaast, enzovoorts.

Ook hier een klein voorbeeld om de opbouw de laten zien:

```
100 SCREEN 2
110 FOR T = 0 TO 1535
120 VPOKE T, &HB1 :' Linker pixel geel, rechter zwart
130 FOR I = 1 TO 50: NEXT
140 NEXT
150 END
```

## SPRITES

Op scherm 1 of 2 is het mogelijk om SPRITES te laten zien, kleine tekeningen (van 1 kleur) waarvan alleen een coördinaat opgegeven hoeft te worden. De VDP zorgt ervoor dat deze tekeningen op de juiste plaats zichtbaar zijn.

Om met de sprites te kunnen werken, zijn er twee tabellen in VRAM: de sprite patroon- en de sprite statutabel.

De sprite patroontabel bevat de tekeningen (8\*8 of 16\*16 bits) en staat op VRAM adressen 14336 tot en met 16383. In het geval van 8\*8 bits (SCREEN ,0 of 1) kan de sprite patroontabel 256 verschillende tekeningen bevatten, waarbij acht opeenvolgende bytes het aanzien van een gehele sprite bepalen. De sprite patroontabel kan 64 verschillende tekeningen bevatten als voor 16\*16 bit sprites gekozen is (SCREEN ,2 of 3). 16 opeenvolgende VRAM bytes bepalen dan het aanzien van de linker helft van een sprite en de 16 daarna volgende bytes doen dat voor de rechter helft.

De sprite statutabel is 128 bytes lang (4 bytes per sprite, maximaal 32 sprites) en staat op VRAM adressen 6912 tot en met 7037. De eerste 4 bytes bepalen de status van sprite 0, de volgende 4 doen dat voor sprite 1, enzovoort.

Per 4 bytes bepaalt het 1e byte de verticale positie, het 2e byte de horizontale positie, het derde byte geeft aan welke sprite tekening gebruikt wordt (van het derde byte worden alleen de linker 5 bits gebruikt als de 16\*16 bit sprites actief zijn) en het 4e byte bepaalt tenslotte de

kleur van de sprite. Tevens bepaalt het meest linker bit (bit 7) van het 4e byte of de gehele sprite al dan niet 32 posities naar links op het scherm geplaatst wordt.

Er kunnen nooit meer dan vier sprites naast elkaar op het beeldscherm weergegeven worden. Mocht dit "gaan" gebeuren dan verdwijnt de sprite met het hoogste nummer geheel of gedeeltelijk.

Een voorbeeld van de opbouw van een 8\*8 bit sprite

```
100 SCREEN 1,1
110 VPOKE 6912, 90: 'Verticaal
120 VPOKE 6913, 110: 'Horizontaal
130 VPOKE 6914, 0: 'Tekening 0
140 VPOKE 6915, 15: 'Wit
150 FOR T = 14336 TO 14343
160 VPOKE T, 255
170 FOR I = 1 TO 100: NEXT
180 NEXT
190 GOTO 190
200 END
```

Een voorbeeld van de opbouw van een 16\*16 bit sprite

```
100 SCREEN 2,3
110 VPOKE 6916, 20: 'Verticaal
120 VPOKE 6917, 30: 'Horizontaal
130 VPOKE 6918, 0: 'Tekening 0
140 VPOKE 6919, 1: 'Zwart
150 FOR T = 14336 TO 14367
160 VPOKE T, 255
170 FOR I = 1 TO 100: NEXT
180 NEXT
190 GOTO 190
200 END
```

\* \* \* \* \*







We kunnen het eindelijk bekend maken (na maanden bezig te zijn geweest met "sleutelen" en testen): er is voor Basicode een geheel nieuw vertaalprogramma. Het draait op de paste versie ook op de .728 (en andere MSX-computers met 64k RAM).

De programmeur spreekt in zijn uitleg (een extra programma op de nieuwe cassette) terecht van een uitvoering die luxer, uitgebreider en gebruikersvriendelijker is. Maar eerlijk gezegd is dat ZEER bescheiden weergegeven.

Het reeds bestaande programma is zeer goed, echter, de ervaringen in de afgelopen maanden opgedaan (o.a. met de pogingen tot ontvangst van BBC1 en de enorm toegenomen deskundigheid van de programmeur) hebben een geheel nieuw produkt opgeleverd dat een summum aan vakbekwaamheid weerspiegelt. We mogen gerust stellen, dat de "meester" zijn vorig "meesterwerk" zeer ver heeft overtroffen.

Het hoofdprogramma (bload"basico",r) start direkt op met een mooi plaatje. Hierna kan men met "J" om de uitleg vragen (dit programma op de band staat achter het hoofdprogramma) of men drukt op "N" en komt in het keuzemenu. Door een toets in te drukken (A t/m Q) geeft dit aan welke delen van het programma, na een druk {ESC<, geactiveerd worden. Ook visueel zeer attractief.

Als men het nieuwe vertaalprogramma eenmaal gebruikt heeft, wil me. niets anders: er is namelijk ook een simpele, maar zeer handige tekstverwerker ingebouwd!! waarmee een niet goed ingeladen Basicode-programma gecorrigeerd kan worden, voordat het vertaald wordt naar SVI/MSX-Basic.

In de info's 4 en 5 spraken wij o.a. over een "Kopfnuss" (hersenkraaker) die via WDRIII wordt uitgezonden in het TV-programma KNOW-HOW. Om te verduidelijken wat het precies is, plaatsen wij onderstaand een leuke opgave uitgezonden op zondag 4 november '84. We maken u erop attent dat er in dit geval geen gebruik hoeft te worden gemaakt van de sub-routines (regels 10-360). Dus laden met mon\* (of schema C van het nieuwe vertaalprogramma).

De regels staan door elkaar en men dient experimenteel de juiste volgorde uit te zoeken.

```

1000 K=L:M=0:IFB<2 THEN1140
1010 INPUTA,B,C: GOSUB1070:F=E:PRINT:G=F-D
1020 PRINT"ZWEITES DATUM (TT,MM,JJJJ) :";
1030 H=K:I=4: GOSUB1150:N=L:IFN{<0 THEN1140
1040 H=C:I=100:J=INT(C/I): GOSUB1150
1050 HOME:REM BILDSCHIRM LOESCHEN
1060 L=H-I*INT(H/I):RETURN
1070 M=2:IFK=0 THEN1120
1080 M=1
1090 INPUTA,B,C: GOSUB1070:D=E
1100 PRINT"ZWISCHEN DIESEN DATEN LIEGEN ";
1110 GOTO1130
1120 PRINTG;:PRINT" TAGE.":END
1130 H=J:I=4: GOSUB1150:IFL{<0 THEN1140
1140 E=INT(365.25*K)+INT(30.56*B)+M+A:RETURN
1150 PRINT"ERSTES DATUM (TT,MM,JJJJ) :";
2000 REM UNSERE ADRESSE :
2010 REM
2020 REM WDR
2030 REM KNOW HOW
2040 REM POSTFACH
2050 REM 5 KOELN 100

```

De oplossing kan men 1 tot 2 dagen na uitzending op de de Teletext van WDR III vinden (Pag. 350 e.v.). Bij deze de oplossing van bovenstaande opgave:

```

1000 HOME:REM BILDSCHIRM LOESCHEN
1010 PRINT"ERSTES DATUM (TT,MM,JJJJ) :";
1020 INPUTA,B,C: GOSUB1070:D=E
1030 PRINT"ZWEITES DATUM (TT,MM,JJJJ) :";
1040 INPUTA,B,C: GOSUB1070:F=E:PRINT:G=F-D
1050 PRINT"ZWISCHEN DIESEN DATEN LIEGEN ";
1060 PRINTG;:PRINT" TAGE.":END
1070 H=C:I=100:J=INT(C/I): GOSUB1150
1080 K=L:M=0:IFB<2 THEN1140
1090 M=2:IFK=0 THEN1120
1100 H=K:I=4: GOSUB1150:N=L:IFN{<0 THEN1140
1110 GOTO1130
1120 H=J:I=4: GOSUB1150:IFL{<0 THEN1140
1130 M=1
1140 E=INT(365.25*K)+INT(30.56*B)+M+A:RETURN
1150 L=H-I*INT(H/I):RETURN
2000 REM UNSERE ADRESSE :
2010 REM
2020 REM WDR
2030 REM KNOW HOW
2040 REM POSTFACH
2050 REM 5 KOELN 100

```

Nog wel in regel 1000 het statement HOME vervangen door CLS. De eerstvolgende uitzending op WDR III is zondag 7 april, 17.30 uur.

tot de volgende keer.